

JACOBI – Eigenvalues of Symmetric Matrices – Jacobi’s Method

© 2020 Valentín Albillo

Abstract

JACOBI is a program written in 1980 for the HP-41C programmable calculator to find all eigenvalues of a real NxN symmetric matrix using Jacobi’s method. One worked example and two solved test cases included.

Keywords: eigenvalues, symmetric matrix, Jacobi’s method, RPN, programmable calculator, HP-41C, HP42S

1. Introduction

JACOBI is a 215-step RPN program that I wrote in 1980 for the *HP-41C* programmable calculator (will also run *as-is* in the *HP-41CV/CX* and in the *HP42S* with trivial changes, see *Note 1*) to obtain numerical values for all eigenvalues of a real NxN symmetric matrix up to 22x22.

It uses *Jacobi’s method*, which annihilates in turn selected off-diagonal elements of the given matrix using elementary orthogonal transformations in an iterative fashion until all off-diagonal elements are 0 when rounded to a user-specified number of decimal places. Once this happens the diagonal elements are the eigenvalues. The method proceeds as follows:

Consider the real NxN symmetric matrix *A*:

$$A = \begin{pmatrix} a_{11} & a_{12} & \dots & a_{1N} \\ a_{12} & a_{22} & \dots & a_{2N} \\ \dots & \dots & \dots & \dots \\ a_{1N} & a_{2N} & \dots & a_{NN} \end{pmatrix}$$

A is symmetric, this is: $a_{ij} = a_{ji}$

We want to find its eigenvalues: a given value *x* is an eigenvalue of a matrix *A* if:

$$A \cdot \bar{e} = x \cdot \bar{e} \quad , \quad \text{where } \bar{e} \text{ is some vector.}$$

This implies that all eigenvalues are the roots of:

$$\begin{vmatrix} a_{11} - x & a_{12} & \dots & a_{1N} \\ a_{12} & a_{22} - x & \dots & a_{2N} \\ \dots & \dots & \dots & \dots \\ a_{1N} & a_{2N} & \dots & a_{NN} - x \end{vmatrix} = 0$$

However, expanding this NxN determinant to get the *characteristic polynomial* and then finding all N real roots of the resulting N-th degree equation, which will be the sought-for matrix eigenvalues, is rather cumbersome and present all kinds of potential difficulties.

The *Jacobi’s method* does not attempt to explicitly generate and solve the equation but proceeds iteratively. It is based on the fact that an NxN symmetric matrix has exactly N real eigenvalues (not necessarily distinct). We can find another NxN matrix *S* so that $S A S^T = D$, where *D* is a *diagonal matrix* whose elements are precisely the eigenvalues of *A*.

The method starts from the original matrix *A* and keeps on annihilating selected off-diagonal elements by performing elementary *rotations*. Let’s single out an off-diagonal element, say *Apq*, and using an elementary rotation we’ll annihilate it. The *rotation matrix R* will be:

$$\begin{aligned} R_{pp} &= \cos Z, & R_{pq} &= \sin Z, & R_{qp} &= -\sin Z, & R_{qq} &= \cos Z \\ R_{ii} &= 1, & R_{pk} &= R_{iq} = R_{ik} = 0 & \text{for } i \neq p, q & \text{ and } k \neq p, q \end{aligned}$$

Let's denote $\mathbf{B} = \mathbf{R}^t \mathbf{A} \mathbf{R}$, whose elements are:

$$\left. \begin{aligned} B_{ip} &= A_{ip} \cos Z - A_{iq} \sin Z \\ B_{iq} &= A_{ip} \sin Z + A_{iq} \cos Z \\ B_{ik} &= A_{ik} \end{aligned} \right\}, \text{ where } i, k \neq p, q$$

$$\begin{aligned} B_{pp} &= A_{pp} \cos^2 Z + A_{qq} \sin^2 Z - 2 A_{pq} \sin Z \cos Z \\ B_{qq} &= A_{pp} \sin^2 Z + A_{qq} \cos^2 Z + 2 A_{pq} \sin Z \cos Z \\ B_{pq} &= 0 \end{aligned}$$

and the remaining elements are symmetric. We also have:

$$\left. \begin{aligned} \sin Z &= \frac{w}{\sqrt{2(1 + \sqrt{1 - w^2})}} \\ \cos Z &= \sqrt{1 - \sin^2 Z} \end{aligned} \right\}, \text{ where } w = \frac{\text{sign}(M) \cdot L}{\sqrt{M^2 + L^2}} \text{ and } L = -A_{pq}, \quad M = \frac{1}{2}(A_{pp} - A_{qq})$$

This is iterated using a strategy for selecting each non-diagonal element in turn until all non-diagonal elements are zero when rounded to a user-specified number of decimal places. Once this happens the matrix diagonal contains the eigenvalues.

1.1 Program characteristics

This program is 46 registers long and can handle symmetric matrices from 2x2 to 22x22, both limits included. If a bare-bones *HP-41C* is used (no memory modules), matrices up to 3x3 may be treated¹. On the other hand, if plugging all 4 memory modules in (or using the *HP-41CV/CX*) matrices up to 22x22 can be dealt with.

All inputs and outputs are labeled and there's a warning to prevent a memory allocation error (**SIZE**).

The accuracy and running times depend on the display setting, however the computed eigenvalues are often more accurate than that: for instance, using **FIX 5** is quite possible to get 7-8 correct digits.

The program has been written to be as fast as possible and to fit into a minimum amount of program memory. For example, the matrix is stored taking into account its symmetry (a_{ij} is stored only for $i \leq j$) and this results in fast execution times and less storage registers needed, as the NxN matrix uses $(N^2+N)/2$ registers instead of N^2 . The minimum **SIZE** to work with an NxN matrix is **SIZE** $(N^2+N) / 2 + 10$.

Running times depend on the display setting as well as the matrix itself. However, the method converges always and is stable against rounding errors. The following times for the original, physical *HP-41C* are indicative:

2x2	3x3	4x4	5x5	6x6
8 sec.	50 sec.	2,5 min.	6 min.	10 min.

¹ Up to 4x4 if using *synthetic instructions*, in which case be careful not to insert a final **END** instruction, let the final **.END**. of program memory do the job instead or 4x4 matrices won't be feasible. See the footnote in the following page.

2. Program Listing¹

01	◆ LBL "JB"	44 X<>Y	87 SIGN	130 XEQ 90 ▶	173 ISG 03	-215 program steps
02	FIX 0	45 ISG 03	88 RCL 06	131 STO 05	174 X<>Y	
03	CF 29	46 GTO 07 ▶	89 LASTX	132 RDN	175 X≠Y?	- uses R ₀₀ –R ₀₉ as
04	"N=?"	47 ISG 02	90 X↑2	133 STO 08	176 GTO 85 ▶	auxiliary registers,
05	PROMPT	48 GTO 12 ▶	91 RCL 06	134 RCL 07	177 FS?C 00	plus the ones used
06	STO 00	49 "FIX=?"	92 X↑2	135 STO 09	178 GTO 70 ▶	to store the matrix
07	ENTER↑	50 PROMPT	93 +	136 *	179 RCL 00	
08	X↑2	51 FIX IND X	94 SQRT	137 RCL 03	180 1 E3	
09	+	52 ◆LBL 70	95 /	138 RCL 04	181 /	- uses flag 00
10	2	53 2	96 *	139 XEQ 90 ▶	182 1	
11	/	54 STO 03	97 1	140 STO T	183 +	- FIX 5 setting is
12	9	55 ◆LBL 85	98 RCL Y	141 RDN	184 STO 06	recommended
13	+	56 1	99 X↑2	142 ST* 09	185 ◆LBL 13	
14	SF 25	57 ◆LBL 87	100 -	143 RCL 06	186 "X="	
15	RCL IND X	58 STO 02	101 SQRT	144 ST* 08	187 RCL 06	
16	FS?C 25	59 CF 00	102 1	145 *	188 INT	
17	GTO 00 ▶	60 RCL 03	103 +	146 -	189 ENTER↑	
18	"SIZE "	61 XEQ 90 ▶	104 ST+ X	147 STO IND 05	190 XEQ 90 ▶	
19	1	62 X<>Y	105 SQRT	148 RCL 08	191 X<>Y	
20	+	63 RND	106 /	149 RCL 09	192 ARCL X	
21	ARCL X	64 X=0?	107 STO 06	150 +	193 PROMPT	
22	PROMPT	65 GTO 84 ▶	108 ST* 01	151 STO IND Z	194 ISG 06	
23	◆LBL 00	66 SF 00	109 X↑2	152 RCL 04	195 GTO 13 ▶	
24	10	67 LASTX	110 ST* 07	153 RCL 00	196 RTN	
25	STO 04	68 ST- IND Z	111 1	154 X≤Y?	197 ◆LBL 90	
26	RCL 00	69 STO 01	112 STO 04	155 GTO 84 ▶	198 X>Y?	
27	1 E3	70 ST+ 01	113 X<>Y	156 ◆LBL 08	199 X<>Y	
28	/	71 CHS	114 -	157 1	200 RCL 00	
29	1	72 STO 06	115 SQRT	158 ST+ 04	201 ST+ X	
30	+	73 RCL 02	116 ST* 01	159 RCL 00	202 X<>Y	
31	STO 02	74 RCL 02	117 X<> 07	160 RCL 04	203 -	
32	◆LBL 12	75 XEQ 90 ▶	118 RCL 01	161 X≤Y?	204 1	
33	RCL 02	76 STO 08	119 +	162 GTO 01 ▶	205 ST- L	
34	STO 03	77 RDN	120 ST- IND 08	163 ◆LBL 84	206 X<> L	
35	◆LBL 07	78 RCL 03	121 ST+ IND 09	164 RCL 03	207 *	
36	"A"	79 RCL 03	122 ◆LBL 01	165 RCL 02	208 2	
37	ARCL 02	80 XEQ 90 ▶	123 RCL 03	166 1	209 /	
38	" , "	81 STO 09	124 RCL 04	167 +	210 +	
39	ARCL 03	82 RDN	125 X=Y?	168 X≠Y?	211 9	
40	" =?"	83 -	126 GTO 08 ▶	169 GTO 87 ▶	212 +	
41	PROMPT	84 STO 07	127 RCL 02	170 X<>Y	213 RCL IND X	
42	STO IND 04	85 2	128 X=Y?	171 RCL 00	214 X<>Y	
43	ISG 04	86 /	129 GTO 08 ▶	172 X≠Y?	215 .END.	

The symbols ◆ and ▶ are purely cosmetic, to indicate branching; | is the **Append** alpha function; = see footnote.

¹ If you're able to use *synthetic instructions* you can save three additional registers for other purposes by using registers **M**, **N** and **O** instead of **R₀₇-R₀₉**, respectively. To that effect, perform the following changes in the program listing above:

- change all instructions using registers **07/08/09** to use instead registers **M/N/O**, respectively.
- change line **12 9** to **12 6**, line **24 10** to **24 7** and line **211 9** to **211 6**
- the size required for NxN matrices will now be **SIZE (N²+N)/2+7** and 4x4 will be possible with no extra **RAM**.

3. Usage Instructions

To find all the eigenvalues of a real NxN symmetric matrix **A**, follow these instructions:

In **RUN** Mode, execute the program:

XEQ "JB" → $N=?$ { asks for the size of the matrix }
N **R/S** → $A_{i,j}=?$ { asks for the matrix elements in turn, from left to right and from top to bottom }
 a_{ij} **R/S** → ... { due to symmetry, the program will only ask for the upper-half elements }
 → $FIX=?$ { asks for the desired number of decimal places (accuracy and run time depend on this value), specifying 5 decimals is recommended }
dec. **R/S** → ... { the program runs until all eigenvalues have been computed, and finally ... }
 ... → $X=(1^{st} \text{ eigenvalue})$
R/S → $X=(2^{nd} \text{ eigenvalue})$ **R/S** → ... **R/S** → $X=(N^{th} \text{ eigenvalue})$

All eigenvalues are displayed rounded to the specified number of decimals. However, each value is also placed without rounding in the X stack register so they can be seen to any number **n** of decimals by using **FIX n**. For another matrix, repeat the procedure above.

Notes:

- if after introducing **N** a message **SIZE nnn** does show up this means the current register allocation is insufficient to run the program so simply execute **SIZE nnn** as directed and then press **R/S** to resume. In general, if the matrix dimension is **N**, a minimum of $(N^2+N)/2+10$ registers are required.
- input of the matrix elements may use the stack but don't disturb the stack while the output is taking place.
- accuracy and running time both depend on the number of digits **d**. As a guideline, use **d = 5** to speed things up.

4. Examples

The following examples can be useful to check that the program is correctly entered and to understand its usage.

4.1 Example 1

Find all the eigenvalues of the following 4x4 matrix. Specify 5 decimal digits.

$$A = \begin{pmatrix} 25 & -41 & 10 & -6 \\ -41 & 68 & -17 & 10 \\ 10 & -17 & 5 & -3 \\ -6 & 10 & -3 & 2 \end{pmatrix}$$

XEQ "JB" → $N=?$
 4 **R/S** → $A_{1,1}=?$ 25 **R/S** → $A_{1,2}=?$ -41 **R/S** → $A_{1,3}=?$ 10 **R/S** → $A_{1,4}=?$ -6
R/S → $A_{2,2}=?$ 68 **R/S** → $A_{2,3}=?$ -17 **R/S** → $A_{2,4}=?$ 10
R/S → $A_{3,3}=?$ 5 **R/S** → $A_{3,4}=?$ -3
R/S → $A_{4,4}=?$ 2
R/S → $FIX=?$ 5 **R/S** → ... { and after 2 min. 22 sec. the eigenvalues are displayed }
R/S → $X=0.03302$
R/S → $X=98.52170$ these are the four eigenvalues; they're correct to more than the digits shown
R/S → $X=1.18609$
R/S → $X=0.25920$

4.2 Test case 1

Compute the eigenvalues of the following 3x3 matrix. Specify 8 decimal digits.

$$A = \begin{pmatrix} 1 & 1 & \frac{1}{2} \\ 1 & 1 & \frac{1}{4} \\ \frac{1}{2} & \frac{1}{4} & 2 \end{pmatrix}$$

The program should give: $X_1 = 2.53652586$, $X_2 = -0.01664728$, $X_3 = 1.48012142$, in about 50 sec.

4.3 Test case 2

Compute the eigenvalues of the following 5x5 matrix. Specify 5 decimal digits.

$$A = \begin{pmatrix} 1 & -2 & 4 & 3 & 6 \\ -2 & 2 & -3 & 0 & -1 \\ 4 & -3 & 3 & 6 & 4 \\ 3 & 0 & 6 & 5 & 2 \\ 6 & -1 & 4 & 2 & -2 \end{pmatrix}$$

The program should give: $X_1 = 0.61259$, $X_2 = 15.39409$, $X_3 = -3.23854$,
 $X_4 = 3.11890$, $X_5 = -6.88703$, in about 5 min. 57 sec.

All values can be tested: simply subtract the eigenvalue from each diagonal element and check that $\text{Det}(A) \sim 0$.

Notes

1. This program can easily be adapted to run under other RPN versions with minimal changes, e.g. to run it on the HP42S simply change the names of instructions such as **FIX 0** to **FIX 00**, * to **x**, **ST+ / ST- / ...** to **STO+ / STO- / ...**, **ARCL X** to **ARCL ST X** and so on. See the *HP42S Owner's Manual* section titled "Using HP-41C Programs" for full details. The size of the resulting program for the HP42S should be 337 bytes.

Also, the HP42S is much faster and has a wider numeric range and greater precision (12-digit instead of 10-digit), so the results might differ slightly. If running the resulting HP42S program on the Free42 emulator (which is thousands of times faster than the original, physical HP-41C) all *Examples* above run almost instantly.

2. This program was published in *PPC Technical Notes VIN3 pp9-13* (October 1980).

References

- Francis Scheid (1988). *Schaum's Outline of Theory and Problems of Numerical Analysis, 2nd Edition*.
Hewlett-Packard (1988). *HP42S Owner's Manual ("Using HP-41C Programs" section), Edition 1 June 1988*.

Copyrights

Copyright for this paper and its contents is retained by the author. Permission to use it for non-profit purposes is granted as long as the contents aren't modified in any way and the copyright is acknowledged.